
Zahlen addieren mit Kritieren: Excel-
Funktion SUMMEWENN()

Sie können mit der Funktion SUMMEWENN() Zahlen summieren, wenn
bestimmte Kriterien erfüllt sind. Die Bedingungen und Kriterien können
Sie flexibel festlegen. Die folgenden Beispiele zeigen die Möglichkeiten.

Zuletzt geändert am 18.03.2026

Die Excel-Funktion SUMMEWENN() wird verwendet, um die Summe bestimmter
Zellen in einem Bereich basierend auf einer Bedingung zu berechnen. Sie ist eine der
wichtigsten Funktionen in der täglichen Excel-Praxis, die Sie unbedingt kennen sollten.

Parameter und Aufbau der Excel-Funktion
SUMMEWENN()

Die Syntax von SUMMEWENN() lautet:

=SUMMEWENN(Bereich; Kriterium; [Summe_Bereich])

– Bereich: Der Bereich, in dem die Bedingung überprüft und nach dem
Kriterium gesucht wird.

– Kriterium: Die Bedingung, die erfüllt sein muss, damit der Wert in den
Bereich einbezogen wird.

– Summe_Bereich (optional): Der Bereich, dessen Werte summiert werden
sollen. Wenn dieser Parameter ausgelassen wird, werden die Werte aus 
Bereich summiert.

Ein Beispiel für den Einsatz von SUMMEWENN()

Schauen Sie sich die Funktionsweise von SUMMEWENN() an einigen praktischen
Beispielen an.

Die folgende Abbildung enthält eine Aufstellung von Produktverkäufen, die als
Ausgangsbeispiel dienen soll.

Datenquelle zur Auswertung mit SUMMEWENN()

Im ersten Beispiel sollen die Umsätze von den Produkten summiert werden, die eine
blaue Farbe haben. Erfassen Sie hierzu die folgende Formel, zum Beispiel in Zelle F15:

=SUMMEWENN(B2:B12; "blau"; F2:F12)

Da das Suchkriterium "blau" im Bereich B2:B12 steht, erfassen Sie diesen als erstes
Argument (Bereich).

Das Suchkriterium "blau" wird anschließend als zweites Argument in der Funktion
angegeben.

Zum Schluss erfassen Sie noch den Summenbereich F2:F12, da hier die Umsätze in
der Tabelle erfasst sind.

Es werden alle Umsätze addiert, bei denen in der Spalte B der Eintrag blau steht.

Wichtig: Suchkriterien in Textform müssen in doppelte Anführungszeichen ("")
gesetzt werden.

Ergebnis der Funktion SUMMEWENN() mit Eingabe eines Kriteriums

Im nächsten Beispiel wollen Sie die Umsätze berechnen, deren Produkte ein Gewicht
von 10 kg haben. Erfassen Sie hierzu die folgende Formel in der Zelle F15:

=SUMMEWENN(C2:C12; 10; F2:F12)

Wichtig: Bei numerischen Suchkriterien (Zahlen) dürfen Sie keine doppelten
Anführungszeichen erfassen, da es sich um keine Texte handelt.

SUMMEWENN() mit numerischem Suchkriterium

SUMMEWENN() mit Platzhalter-Kriterien

Sie können auch die bekannten Wildcards oder Platzhalter wie * und ? in den
Suchkriterien einsetzen:

– ? zur Übereinstimmung mit einem beliebigen einzelnen Zeichen

– * zur Übereinstimmung mit einer beliebigen einzelnen Zeichenfolge

Wollen Sie die Summe der grünen und gelben Umsätze berechnen, dann können Sie
das Sternchen als Wildcard wie folgt einsetzen:

=SUMMEWENN(B2:B12; "g*"; F2:F12)

Wichtig: Wenn Sie nach einem Fragezeichen oder Sternchen selbst suchen möchten,
dann müssen Sie eine Tilde (~) vor dem zu suchenden Zeichen erfassen.

Mit SUMMEWENN() und Platzhaltern suchen und summieren

SUMMEWENN() mit Größenvergleich als
Auswahlkriterium

Sie können auch die bekannten Vergleichsoperatoren < = > in den Suchkriterien
einsetzen. Wollen Sie zum Beispiel die Umsätze von den Produkten addieren, deren
Verkaufspreis größer oder gleich 180,00 ist, dann verwenden Sie die folgende Formel:

=SUMMEWENN(E2:E12; ">=180"; F2:F12)

Wichtig: Beim Einsatz von Vergleichsoperatoren müssen Sie das Suchkriterium in
doppelte Anführungszeichen (") setzen.

SUMMEWENN() mit Vergleichsoperatoren

Wenn Sie einen Vergleichsoperator < = > mit einem Verweis auf eine andere Zelle als
Suchkriterium in SUMMEWENN() einsetzen wollen, dann müssen Sie

– den Vergleichsoperator in Anführungszeichen ("") erfassen,

– dann ein &-Zeichen einfügen und

– dahinter den Zellverweis nennen.

Beispiel: In Zelle D1 steht eine Zahl und Sie wollen den Bereich E2:E12 durchsuchen
und die Werte im Bereich F2:F12 summieren, wenn das Auswahlkriterium erfüllt ist.
Die Formel lautet:

=SUMMEWENN(E2:E12; ">="&D1; F2:F12)

SUMMEWENNS() – Addition und
Wahrheitsabfrage mit Excel

So können Sie flexibel Zahlen addieren, wenn bestimmte Kriterien erfüllt
sind. Mit der Funktion SUMMEWENNS() können Sie bis zu 127 Bereiche
nach den von Ihnen definierten Kriterien prüfen – und dann die
jeweiligen Zahlen addieren.

Zuletzt geändert am 18.03.2026

Die Excel-Funktion SUMMEWENNS() ist eine Weiterentwicklung von SUMMEWENN().
Mit SUMMEWENNS() können Sie die Summe bestimmter Zellen in einem Bereich
basierend auf mehreren Bedingungen berechnen.

Die Parameter der Excel-Funktion SUMMEWENNS()

Die Syntax von SUMMEWENNS() lautet wie folgt:

=SUMMEWENNS(Summe_Bereich; Kriterienbereich1; Kriterium1;
[Kriterienbereich2]; [Kriterium2] …)

– Summe_Bereich: Der Bereich, dessen Werte summiert werden sollen, wenn
die folgenden Kriterien alle zutreffen.

– Kriterienbereich1: Der Bereich, der auf Kriterium1 getestet wird.

– Kriterium1: Die Bedingung, die erfüllt sein muss, damit der Wert in
Summe_Bereich einbezogen wird.

– Kriterienbereich2 (optional): Ein zweiter Kriterienbereich, der durchsucht
wird.

– Kriterien2 (optional): Zusätzliches Kriterium, nach dem im Kriterienbereich2
gesucht wird.

Hinweis: Sie können bis zu 127 Bereich-Kriterien-Paare eingeben.

Seite 8



Kapitel 315 Spezielle Funktionen und Berechnungen in Excel
SUMMEWENNS() – Addition und Wahrheitsabfrage mit Excel

Beispiele für die Funktion SUMMEWENNS()

Einige Beispiele sollen die Funktionsweise von SUMMEWENNS() illustrieren. Die
folgende Abbildung enthält eine Aufstellung von Produktverkäufen, die als Datenquelle
für die Auswahl und die Summierung dienen soll.

Liste mit zu addierenden Werten (Beispiel)

Im ersten Beispiel sollen die Umsätze von den Produkten summiert werden, die eine
grüne Farbe haben. Erfassen Sie hierzu die folgende Formel in die gewünschte
Ergebniszelle; zum Beispiel in der Zelle F15:

=SUMMEWENNS(F2:F12; B2:B12; "grün")

Im ersten Argument erfassen sie den Summenbereich F2:F12, da hier die Umsätze in
der Tabelle eingetragen sind.

Es werden alle Umsätze addiert, bei denen in der Spalte B der Eintrag grün erfasst ist.
Da das Suchkriterium im Bereich B2:B12 steht, definieren Sie diesen als zweites
Argument (Kriterienbereich1) der Funktion.

Das Suchkriterium grün wird anschließend als drittes Argument der Funktion
angegeben.

Wichtig: Suchkriterien in Textform (grün) müssen in doppelte Anführungszeichen ("")
gesetzt werden.

SUMMEWENNS() mit einem Kriterium

SUMMEWENNS() mit mehreren Kriterien zur
Prüfung

Im nächsten Beispiel sollen die Umsätze berechnet werden, deren Produkte grün sind
und ein Gewicht von 10 kg haben. Erfassen Sie hierzu die folgende Formel in Zelle F15:

=SUMMEWENNS(F2:F12; B2:B12; "grün"; C2:C12; 10)

Hier kommt jetzt die große Weiterentwicklung von SUMMEWENNS() im Vergleich zu
SUMMEWENN() zum Vorschein. Sie können einfach ein weiteres Kriterienpaar
(Kriterienbereich und Suchkriterium) als zusätzliche Argumente anhängen.

Das zusätzliche Suchkriterium ist in der Spalte C enthalten. Legen Sie daher den
Bereich C2:C12 als Argument für den Kriterienbereich fest und dann anschließend den
Wert 10 als das Suchkriterium.

Wichtig: Bei numerischen Suchkriterien (im Beispiel die Zahl 10) dürfen Sie keine
doppelten Anführungszeichen eingeben, da es sich um keinen Text handelt.

SUMMEWENNS() mit zwei Kriterien

Weitere Operatoren bei der Funktion
SUMMEWENNS()

Sie können auch die bekannten Vergleichsoperatoren < = > in den Suchkriterien
einsetzen. Wollen Sie zum Beispiel die Umsätze der grünen Produkte addieren, deren
Gewicht größer als 10 kg ist, dann verwenden Sie die folgende Formel:

=SUMMEWENNS(F2:F12; B2:B12; "grün"; C2:C12; ">10")

Wichtig: Beim Einsatz von Vergleichsoperatoren müssen Sie das Suchkriterium in
doppelte Anführungszeichen ("") setzen.

Die Funktion SUMMEWENNS() mit Vergleichsoperator

Außerdem können die bekannten Wildcards wie * und ? in den Suchkriterien einsetzen:

– ? zur Übereinstimmung mit einem beliebigen einzelnen Zeichen

– * zur Übereinstimmung mit einer beliebigen einzelnen Zeichenfolge

Wollen Sie die Summe der Umsätze mit grünen und gelben Produkten berechnen,
deren Gewicht größer als 10 kg ist, dann können Sie das Sternchen als Wildcard wie
folgt einsetzen:

=SUMMEWENNS(F2:F12; B2:B12; "g*"; C2:C12; ">10")

Wichtig: Wenn Sie nach einem Fragezeichen oder Sternchen selbst suchen möchten,
dann müssen Sie eine Tilde (~) vor dem zu suchenden Zeichen erfassen.

Funktion SUMMEWENNS() mit Wildcards

Wenn Sie einen Vergleichsoperator < = > mit einem Verweis auf eine andere Zelle als
Suchkriterium in SUMMEWENNS() einsetzen wollen, dann müssen Sie

– den Vergleichsoperator in Anführungszeichen ("") erfassen,

– dann ein &-Zeichen einfügen und

– dahinter den Zellverweis nennen.

Beispiel: In Zelle G1 steht ein Gewicht und Sie wollen den Bereich C2:C12
durchsuchen und die Werte im Bereich F2:F12 summieren, wenn das Auswahlkriterium
erfüllt ist. Außerdem sollen nur die Umsätze mit grünen Produkten addiert werden. Die
Formel lautet:

=SUMMEWENNS(F2:F12; B2:B12; "grün"; C2:C12; ">="&G1)

Die Beispiele zeigen, wie Sie mit der Funktion SUMMEWENNS() umfangreiche Listen
durchsuchen und dann flexibel Summen bilden.

– Sie durchsuchen die Liste zeilenweise nach den von Ihnen festgelegten
Kriterien.

– Es werden nur die Werte aus den Zeilen addiert, für die alle definierten
Kriterien zutreffen; es wird also eine Und-Prüfung durchgeführt.

Mehrere Zahlen aus einer Excel-Zelle
addieren

Sie können mehrere Zahlen aus einer Zelle auch dann addieren, wenn
die Zahlen in der Zelle durch Zeilenumbruch getrennt sind. Dazu
braucht es aber eine spezielle Formel.

Zuletzt geändert am 18.03.2026

Wenn Sie Daten aus Fremdprogrammen in Excel importieren, kann es vorkommen,
dass dabei mehrere Zahlen in einer Zelle stehen.

Haben Sie in einer Zelle mehrere Zahlen durch einen Zeilenumbruch (Alt + Enter)
getrennt erfasst, dann können Sie diese nicht addieren. Mit diesen Zahlen können Sie
normalerweise keine weiteren Berechnungen durchführen.

Mit der folgenden Formel können auch Zahlen in einer Zelle addiert werden, die durch
Zeilenumbruch getrennt in einer Zelle erfasst worden sind:

=SUMMENPRODUKT(GLÄTTEN(TEIL(WECHSELN(A1; ZEICHEN(10);
WIEDERHOLEN(" "; 999)); ZEILE(INDIREKT("1:"&LÄNGE(A1) -
LÄNGE(WECHSELN(A1; ZEICHEN(10); )) + 1)) * 999-998; 999)) * 1)

Die Formel geht davon aus, dass sich die erfassten Zahlen in der Zelle A1 befinden.

Mehrere Zahlen aus einer Excel-Zelle addieren

Quersumme einer Zahl mit Excel
berechnen

Die Quersumme einer Zahl wird beispielsweise verwendet, um zu
prüfen, ob ein Eintrag korrekt ist. Mit einer Matrixformel können Sie die
Quersumme einer beliebigen Zahl in einer Excel-Zelle berechnen.

Zuletzt geändert am 18.03.2026

Die Quersumme einer Zahl ist die Summe ihrer einzelnen Ziffern. Um die Quersumme
einer Zahl zu berechnen, müssen Sie die Einer, Zehner, Hunderter etc. addieren.

Beispiel: Die Quersumme der Zahl 12345 ist 15 (1+2+3+4+5=15).

Quersummen können Sie zum Beispiel einsetzen, um Prüfverfahren von Daten zu
gewährleisten oder zu integrieren.

Für das Berechnen einer Quersumme gibt es in Excel keine Standardfunktion. Mit der
folgenden Matrixformel können Sie die Quersumme einer Zahl berechnen.

{=SUMME(TEIL(A1; ZEILE(INDIREKT("1:"&LÄNGE(A1))); 1) * 1)}

Die Matrixformel geht davon aus, dass sich die Zahl, für welche die Quersumme
berechnet werden soll, in der Zelle A1 befindet.

Wichtig: Bestätigen Sie die Eingabe der Formel nicht mit Enter, sondern mit der
Tastenkombination Strg + Umschalttaste + Enter, da es sich hierbei um eine
Matrixformel handelt. Die Formel wird durch diese Tastenkombination automatisch in
geschweifte Klammern {} eingerahmt.

Matrixformel für die Berechnung der Quersumme

So funktioniert die Formel:

– Über die Funktion LÄNGE() wird die Anzahl der Ziffern in der Zelle A1
ermittelt.

– Anschließend wird der Reihe nach jede einzelne Ziffer aus der Zelle A1 mit der
Funktion TEIL() separiert und an die Funktion SUMME() übergeben.

– Mit SUMME() wird dann die Quersumme berechnet.

Quersumme eines Datums mit Excel
berechnen

Das Datum ist in einer Excel-Zelle im Format TT.MM.JJJJ eingegeben. Um
mit diesem Format und aus den Ziffern die Quersumme zu berechnen,
braucht es eine spezielle Formel.

Zuletzt geändert am 18.03.2026

Für Datumswerte können Sie Quersummen in Excel berechnen lassen, wie für andere
Zahlen auch.

Die Quersumme des Datums 01.01.2022 ist zum Beispiel 8 (0+1+0+1+2+0+2+2=8).

Mit der folgenden Formel können Sie in einer Excel-Tabelle die Quersumme eines
Datums berechnen.

{=SUMMENPRODUKT(TEIL(TEXT(A1; "TTMMJJJJ"); ZEILE(1:8); 1) * 1)}

Die Formel geht hierbei davon aus, dass sich das Datum, für das die Quersumme
berechnet werden soll, in der Zelle A1 befindet.

Wichtig: Bestätigen Sie die Eingabe der Formel nicht mit Enter, sondern mit der
Tastenkombination Strg + Umschalttaste + Enter, da es sich hierbei um eine
Matrixformel handelt. Die Formel wird durch diese Tastenkombination automatisch in
geschweifte Klammern {} eingerahmt.

Quersumme eines Datums mit Excel berechnen

Summe trotz Fehler – Fehlerwert in Excel
elegant umgehen

Viele Excel-Funktionen wie SUMME() können mit Fehlerwerten nicht
umgehen. Das Ergebnis ist ebenfalls ein Fehlerwert. Mit einer
Matrixformel oder der Funktion AGGREGAT() umgehen Sie dieses
Problem auf elegante Art.

Zuletzt geändert am 18.03.2026

Werte zum Summieren mit Fehlerwerten

Wollen Sie in einer Excel-Tabelle Zellen summieren, in denen teilweise Fehlerwerte
(zum Beispiel #DIV/0!) vorkommen? Dann bekommen Sie als Ergebnis ebenfalls eine
Fehlermeldung, wenn Sie die „normalen“ Wege zum Summieren in Excel wählen.

In der folgenden Abbildung sehen Sie einen Bereich von Werten, die summiert werden
sollen. Da hier Fehlerwerte enthalten sind, bekommen Sie über die Funktion SUMME()
ebenfalls einen Fehlerwert angezeigt.

SUMME() kann Fehlerwerte im Argument nicht bearbeiten

Formel fürs Summieren mit Fehlerwerten

Wollen Sie einen solchen Bereich mit Daten trotzdem summieren und die Zellen mit
den Fehlerwerten ignorieren, dann können Sie hierfür die folgende Formel verwenden:

{=SUMME(WENN(ISTFEHLER(A2:C11); 0; A2:C11))}

Hinweis: Erfassen Sie die Formel ohne die geschweiften Klammern am Anfang und am
Ende. Bestätigen Sie die Formel dann aber nicht mit Enter, sondern mit der

Tastenkombination Strg + Umschalttaste + Enter, da es sich um eine Matrixformel
handelt. Durch die Tastenkombination werden dann automatisch die geschweiften
Klammern am Anfang und Ende eingefügt. Nutzen Sie ein Microsoft-365-Abo, dann
können Sie die Formel auch normal mit Enter erfassen.

Werte mit Fehlern summieren

Wie funktioniert die Summenformel mit
Fehlerwerten?

Zunächst wird mit der Funktion ISTFEHLER(A2:C11) jede Zelle im Bereich, der
addiert werden soll (hier: A2:C11), untersucht, ob ein Fehlerwert vorliegt oder nicht.

Liegt ein Fehlerwert vor, dann wird über die WENN()-Funktion der Wert 0, sonst der
tatsächliche Wert der Zelle zurückgegeben: =SUMME(WENN(ISTFEHLER(A2:C11); 0;
A2:C11))

Seite 21



Kapitel 315 Spezielle Funktionen und Berechnungen in Excel
Summe trotz Fehler – Fehlerwert in Excel elegant umgehen

Die über die Funktion WENN() zurückgegebenen Werte werden anschließend an die
Funktion SUMME() weitergeleitet und von dieser summiert:
=SUMME(WENN(ISTFEHLER(A2:C11); 0; A2:C11))

Auf diese Weise gehen nur die Zellen ohne Fehlerwerte in die Summierung mit ein.

Alternative Funktion AGGREGAT()

Eine Alternative für diese Aufgabenstellung ist die Funktion AGGREGAT() in Excel. Sie
ermöglicht es, Berechnungen wie zum Beispiel Mittelwert, Summe oder Maximum
durchzuführen, während bestimmte Werte oder Fehler ignoriert werden können.

Das ist besonders nützlich, wenn man Daten analysiert, die Ausreißer oder Fehler
enthalten.

Nutzen Sie die folgende Funktion, um die Daten im Bereich A2:C11 ohne die
Fehlerwerte zu summieren:

=AGGREGAT(9; 6; A2:C11)

Summieren von Werten mit Fehlerwerten mit der Excel-Funktion AGGREGAT()

Die Funktion AGGREGAT() in Excel ist äußerst flexibel und leistungsstark, da sie
mehrere Berechnungsfunktionen integriert.

Außerdem lassen sich bestimmte Werte oder Fehler in den Daten ignorieren. Dies ist
besonders hilfreich in umfangreichen Datensätzen, bei denen man zum Beispiel
Nullwerte, Fehler oder ausgeblendete Zellen nicht berücksichtigen möchte.

AGGREGAT() besitzt die folgende Syntax:

=AGGREGAT(Funktionsnummer; Optionen; Bereich)

Funktionsnummer
Eine Zahl zwischen 1 und 19, die die auszuführende Berechnung angibt.

1. MITTELWERT

2. ANZAHL

3. ANZAHL2

4. MAX

5. MIN

6. PRODUK

7. STABW.S

8. STABW.N

9. SUMME

10. VAR.S

11. VAR.P

12. MEDIAN

13. MODUS.EINF

14. KGRÖSSTE

15. KKLEINSTE

16. QUANTIL.INKL

17. QUARTILE.INKL

18. QUANTIL.EXKL

19. QUARTILE.EXKL

Im Beispiel muss die Funktionsnummer 9 für Summe erfasst werden.

Optionen
Eine Zahl zwischen 0 und 7, die bestimmt, welche Werte ignoriert werden sollen:

0 oder nicht angegeben – Geschachtelte TEILERGEBNIS- und AGGREGAT-Funktionen
ignorieren

1. Ausgeblendete Zeilen, geschachtelte TEILERGEBNIS- und AGGREGAT-
Funktionen ignorieren

2. Fehlerwerte, geschachtelte TEILERGEBNIS- und AGGREGAT-Funktionen
ignorieren

3. Ausgeblendete Zeilen, Fehlerwerte, geschachtelte TEILERGEBNIS- und
AGGREGAT-Funktionen ignorieren

4. Nichts ignorieren

5. Ausgeblendete Zeilen ignorieren

6. Fehlerwerte ignorieren

7. Ausgeblendete Zeilen und Fehlerwerte ignorieren

Im Beispiel muss die Option 6 für Fehlerwerteignorieren angegeben werden.

Bereich
Hier erfassen Sie den Bereich, in dem die zu summierenden Werte stehen. Im Beispiel
ist dies der Bereich A2:C11.

Zahlen in Excel auf volle 5er runden

Wie runden Sie zum Beispiel 5,33834 auf volle 5er nach der zweiten
Stelle, sodass 5,35 angezeigt wird? In diesem Excel-Tipp zeigen wir
Ihnen, wie Sie entsprechend runden.

Zuletzt geändert am 18.03.2026

Mit den Standardfunktionen RUNDEN(), ABRUNDEN() und AUFRUNDEN() können
Sie jede Zahl auf eine bestimmte Anzahl von Stellen runden.

So arbeitet die Funktion RUNDEN()

Mit der Funktion RUNDEN() runden Sie eine Zahl auf eine angegebene Anzahl von
Stellen. Die Funktion besitzt die folgende Syntax:

=RUNDEN(Zahl; Anzahl_Stellen)

Die Argumente bedeuten:

– Zahl: Die Zahl, die gerundet werden soll.

– Anzahl_Stellen: Die Anzahl der Dezimalstellen, auf die Sie die Zahl runden
möchten.

Wollen Sie zum Beispiel die Zahl 5,33834 auf zwei Dezimalstellen (zwei
Nachkommastellen) runden, führt Sie die folgende Formel zum Ziel:

=RUNDEN(A2; 2)

In diesem Beispiel wird davon ausgegangen, dass sich die Zahl 5,33834 in der Zelle A2
befindet.

Zahlen runden

Auf ganze Zahlen runden

Geben Sie für Anzahl_Stellen eine Null an, wird Zahl auf eine Ganzzahl gerundet. Die
Funktion GANZZAHL() schneidet dagegen die Dezimalstellen einfach ab.

Um den links vom Komma stehenden Teil von Zahl, also die Stellen vor dem
Komma, zu runden, verwenden Sie einen negativen Wert für Anzahl_Stellen. Mit
dieser Formel runden Sie die Zahl 3456,28 beispielsweise auf 3460:

=RUNDEN(3456,28; -1) 

Mit der Formel =RUNDEN(3456,28; -2) runden Sie die Zahl auf 3500.

Zahlen auf volle 5er runden

Nun wollen Sie Zahlen auf volle 5er auf zwei Nachkommastellen runden. Hierfür
erweitern Sie die Formel wie folgt:

=RUNDEN(A2/5; 2) * 5

Durch das Dividieren der zu rundenden Zahl (Zelle A2) durch 5 wird die ursprüngliche
Zahl 5,33834 in den Wert 1,067668 umgerechnet.

Dieser Wert wird als Argument Zahl der Funktion RUNDEN() übergeben. Die Funktion
RUNDEN() rundet diesen Wert jetzt auf zwei Nachkommastellen und erhält somit als
Ergebnis 1,07.

Danach multiplizieren Sie das Ergebnis der Funktion RUNDEN() wieder mit dem
ursprünglichen Divisor des Arguments Zahl – im Beispiel 5. Sie erhalten somit das
richtige, auf volle 5er gerundete Ergebnis 5,35.

Runden auf volle 5er mit der Excel-Funktion RUNDEN()

Preis aufrunden mit Excel auf nächste ,99

Wie Sie mit einer einfachen Formel sicherstellen, dass Preise
automatisch auf die nächste ,99 gerundet werden. So automatisieren
Sie Ihre Preisgestaltung in Excel.

Zuletzt geändert am 18.03.2026

Im Alltag begegnen uns häufig Preise, die auf ,99 enden. Ob im Supermarkt, beim
Online-Shopping oder im Einzelhandel – die sogenannte „Psychologische
Preisgestaltung“ ist allgegenwärtig.

Diese Methode zielt darauf ab, Produkte attraktiver erscheinen zu lassen, indem der
Preis knapp unter einer runden Zahl angesetzt wird, wie etwa 9,99 statt 10,00 EUR.

Doch wie lässt sich dieses Preismuster einfach in Excel umsetzen?

Formel für das Aufrunden auf einen ,99-Betrag

Mit einer geschickten Anwendung der Excel-Funktion AUFRUNDEN() können Sie jede
Zahl nach oben korrigieren – zum Beispiel von 0,55 EUR auf 0,99 EUR oder von
2,45 EUR auf 2,99 EUR. Ideal für alle, die ihre Produktpreise mit nur einer Formel
anpassen möchten, ohne manuelle Eingriffe.

Für diese Aufgabenstellung können Sie die folgende Formel verwenden:

=AUFRUNDEN(A1+0,01; 0) - 0,01

In der Zelle A1 befindet sich die auf die nächste zu rundende ,99-Zahl.

Funktion AUFRUNDEN() für die Berechnung der ,99-Preisschwelle

Mit der Funktion AUFRUNDEN() runden Sie die Zahl in der Zelle A1 auf die nächste
ganze Zahl auf. Aus den folgenden beispielhaften Ausgangswerten würden somit die
folgenden Werte:

– aus 1,2 wird 2

– aus 1,8 wird 2

– aus 0,69 wird 1

– aus 2 wird 2

Anschließend wird der aufgerundete Wert um 0,01 subtrahiert, sodass sich der
gewünschte, auf ,99 gerundete Wert ergibt.

Eine Ausnahme stellen hier ganze Zahlen dar. Ist der Ausgangswert zum Beispiel 2,
dann würde sich grundsätzlich nach der Aufrundung der Wert 2 ergeben, sodass sich
als Endergebnis der Wert 1,99 ergeben würde.

Deswegen wird der Wert in der Zelle A1 noch vor dem Aufrundungsvorgang um 0,01
erhöht.

=AUFRUNDEN(A1+0,01; 0) - 0,01

Es ergibt sich somit nach dem Aufrunden der Wert 3, der dann um 0,01 reduziert wird,
sodass Sie als Ergebnis den Wert 2,99 erhalten.

Varianten des Rundens auf Preisschwellen

Besteht das Ziel darin, einen Preis auf den nächsten Wert zu runden, der auf 0,99
endet, dann können Sie die folgende Formel einsetzen:

=RUNDEN(A1; 0) - 0,01

Funktion RUNDEN() für die Berechnung der nächstgelegenen ,99-Preisschwelle

Bei dieser Formel wird auf die nächsten ,99 gerundet. Im Gegensatz zur ersten Formel
wird hierbei auch abgerundet.

Wollen Sie eine andere Preisschwelle berechnen, passen Sie die erste Formel einfach
an. Oft werden auch ,95-Preisschwellen genutzt. Dafür lautet die entsprechende
Formel:

=AUFRUNDEN(A1+0,05; 0) - 0,05

Lesen Sie mehr zu den Hintergründen der psychologischen
Preiswahrnehmung und zur Preisgestaltung.

Weitere Excel-Tools für die Preiskalkulation finden Sie in dieser Anleitung: 
Selbstkosten und Produktpreis berechnen.

Mit der Excel-Funktion OBERGRENZE()
Zahlen auf das nächste Vielfache
aufrunden

Mit den Excel-Formeln OBERGRENZE() oder AUFRUNDEN() berechnen
Sie das nächste Vielfache einer Zahl. Damit können Sie beispielsweise
Ihre Bestellmenge berechnen, wenn der Lieferant eine
Mindestbestellmenge vorgibt.

Zuletzt geändert am 18.03.2026

Beispiel für die Excel-Funktion OBERGRENZE()

Wenn Sie Produkte oder Artikel bei Lieferanten bestellen, sind diese nicht immer in der
tatsächlich benötigten Menge bestellbar. Artikel wie Schrauben sind meist in festen
Größenordnungen verpackt und Sie können nur ganze Päckchen beim Händler
erwerben.

Wie können Sie bei Ihrer Bestellung solche Rahmenbedingungen berücksichtigen und
Ihre Bestellmengen in Excel berechnen?

Bestellmenge mit der Funktion OBERGRENZE()
berechnen

Für derartige Fragestellungen gibt es in Excel die Funktion OBERGRENZE(), mit der
Sie auf das nächste Vielfache eines angegebenen Wertes aufrunden können. Die
Funktion OBERGRENZE() besitzt die folgende Syntax:

=OBERGRENZE(Zahl; Schritt)

– Zahl ist der Wert, den Sie runden möchten, im Beispiel Ihr tatsächlicher
Bedarf an Schrauben

– Schritt entspricht dem Vielfachen, auf das Sie runden wollen; das ist im
Beispiel die Mindestbestellmenge

Für das Beispiel bedeutet dies:

– Erfassen Sie die Mindestbestellmengen des Händlers in der Spalte A und

– die tatsächlich benötigten Mengen in der Spalte B eines Tabellenblattes.

Dann können Sie in der Spalte C die Bestellmengen wie folgt mit der Funktion
OBERGRENZE() kalkulieren:

=OBERGRENZE(B2; A2)

Bestellmengen mit Excel-Funktion OBERGRENZE() berechnen

Hinweis: Ab Excel 2013 können Sie auch die Funktion 
OBERGRENZE.MATHEMATIK() verwenden, die grundsätzlich identisch zur Funktion
OBERGRENZE() ist. OBERGRENZE.MATHEMATIK() weist aber noch das optionale
Argument Modus auf, mit welchem Sie für eine negative Zahl steuern können, ob diese
in Richtung des größeren oder des kleineren Werts gerundet werden soll.

Bestellmenge mit der Funktion AUFRUNDEN()
berechnen

Die Bestellmenge können Sie auch mit folgender Formel berechnen:

=AUFRUNDEN(B2/A2; 0) * A2

Die Formel geht davon aus, dass sich die benötigte Menge in der Zelle B2 und die
Mindestbestellmenge (Päckchengröße) in Zelle A2 befindet.

Im ersten Teil der Formel =AUFRUNDEN(B2/A2; 0) * A2 wird die Anzahl der
benötigten Päckchen mit der Funktion AUFRUNDEN() berechnet.

Die Funktion AUFRUNDEN(Zahl; Anzahl_Stellen) besitzt die folgende Syntax:

– Zahl ist der Wert, den Sie aufrunden möchten; im Beispiel ist dies der Wert
des Quotienten aus benötigter Menge und Mindestbestellmenge (B2/A2)

– Anzahl_Stellen gibt an, auf wie viele Dezimalstellen die Zahl gerundet
werden soll; dies entspricht hier dem Wert 0, da eine ganze Zahl als
Zwischenergebnis benötigt wird, weil nur ganze Päckchen ausgeliefert werden

Sie erhalten bei einer Mindestbestellmenge von 75 und einer tatsächlich benötigten
Menge von 80 zunächst den Wert 1,07. Dieser Wert ergibt aufgerundet auf
0 Dezimalstellen die Zahl 2.

Das Zwischenergebnis 2 wird anschließend im zweiten Teil der Formel
=AUFRUNDEN(B2/A2; 0) * A2 wieder mit der Mindestbestellmenge multipliziert, sodass
Sie als Ergebnis die Bestellmenge 150 erhalten.

Bestellmengen mit Excel-Funktion AUFRUNDEN() berechnen

Weiteres Beispiel für das Aufrunden auf ein
Vielfaches

Im gewählten Beispiel (oben) geht es um das Aufrunden von Bestellmengen. Die
Funktion OBERGRENZE() können Sie auch für Losgrößen, Lagermengen und andere
Anwendungsfälle nutzen. Zum Beispiel für das Aufrunden von Zeitangaben.

Wenn Sie beispielsweise die tatsächliche Arbeitszeit auf eine volle Stunde, auf halbe
Stunden oder Viertelstunden aufrunden wollen, können Sie dies mit OBERGRENZE()
leicht berechnen.

Sie müssen nur berücksichtigen, dass Excel die Stundenzahl als Anteil des Werts 1
betrachtet. Ein halber Tag (12 Stunden) entspricht deshalb dem Dezimalwert 0,5.

Beispiel: In Zelle A3 steht der Wert der Stunden, die Sie auf halbe Stunden aufrunden
wollen. Sie runden den Wert mit dieser Formel auf:

=OBERGRENZE(A3; 1/48)

Passen Sie noch das Zahlenformat an, da Excel das Ergebnis wieder als Bruchteil von 1
angibt.

OBERGRENZE() für das Aufrunden von Uhrzeiten

Für das Aufrunden auf Viertelstunden lautet die Formel:

=OBERGRENZE(A3; 1/96)

Und für das Aufrunden auf volle Stunden:

=OBERGRENZE(A3; 1/24)

Wie Sie Liefertermine in Excel berechnen

Mit der Funktion ARBEITSTAG() berechnen Sie mit Excel Liefertermine
oder Versandtermine. Dabei berücksichtigen Sie Wochenenden und
Feiertage. Die folgenden Beispiele zeigen, wie Sie die Formel einsetzen.

Zuletzt geändert am 18.03.2026

Die Berechnung von Lieferterminen ist eine häufige Aufgabe in der Logistik, im
Projektmanagement oder bei der Planung von Arbeitsprozessen.

Mit Excel lassen sich diese Berechnungen schnell und einfach automatisieren – auch
wenn Sie Werktage, Wochenenden oder Feiertage berücksichtigen möchten.

In diesem Beitrag erfahren Sie, wie Sie Schritt für Schritt Liefertermine berechnen und
damit immer im Blick haben.

Einfache Berechnung eines Liefertermins

Wenn keine speziellen Bedingungen (zum Beispiel Feiertage oder Wochenenden)
berücksichtigt werden müssen, können Sie einfach das Versanddatum und die
Lieferdauer addieren.

Beispiel:

– Versanddatum: 01.12.2024 (in Zelle A1)

– Lieferzeit: 5 Tage (in Zelle B1)

Die Formel für den Liefertermin lautet:

=A1+B1

Das Ergebnis zeigt den Liefertermin: 06.12.2024.

Beachten Sie dabei, dass diese Addition einen der beiden Tage in die Rechnung
einbezieht. Sollen die beiden Tage (01.12.2024 und 06.12.2024) in der Lieferzeit nicht
berücksichtigt werden, addieren Sie zum Ergebnis 1:

=A1+B1+1

Zählen beide Tage zur Lieferzeit, subtrahieren Sie 1:

=A1+B1-1

Einfache Berechnung von Lieferterminen durch Addition

Wichtig ist, dass Sie sich das Ergebnis der Berechnung im Zahlenformat Datum
anzeigen lassen, damit Sie das Datum ablesen können.

Sollte ein anderes Zahlenformat eingestellt sein, dann können Sie dies schnell ändern,
indem Sie die entsprechende Zelle markieren und anschließend die Tastenkombination
Strg + 1 drücken.

Es öffnet sich das Dialogfeld Zellen formatieren. Klicken Sie hier auf das Register 
Zahlen.

Unter Kategorie wählen Sie Datum aus.

Anschließend können Sie unter Typ das gewünschte Datumsformat markieren.
Bestätigen Sie Ihre Einstellung, indem Sie das Dialogfeld durch einen Klick auf OK
schließen.

Zellwert als Datum anzeigen

Liefertermine mit Berücksichtigung von
Wochenenden

Wenn Sie möchten, dass Wochenenden bei der Berechnung nicht berücksichtigt und
diese Tage ausgeschlossen werden, verwenden Sie die Funktion ARBEITSTAG().

Die ARBEITSTAG()-Funktion in Excel wird verwendet, um ein Datum zu berechnen, das
eine bestimmte Anzahl von Arbeitstagen (ohne Wochenenden) vor oder nach einem
Startdatum liegt. Die Formel lautet:

=ARBEITSTAG(A1; B1)

– A1 ist das Versanddatum.

– B1 ist die Anzahl der Arbeitstage (Lieferdauer).

Hinweis: Die Funktion ARBEITSTAG() betrachtet Samstag und Sonntag als
Wochenende. Sie werden bei der Berechnung in der Form berücksichtigt, dass sich
Lieferzeiten entsprechend verlängern.

Beispiel:

– Versanddatum: 01.12.2024 (ein Sonntag)

– Lieferzeit: 10 Tage

– Das Ergebnis ist der 13.12.2024 (ein Freitag).

Liefertermine berechnen mit Berücksichtigung von Wochenenden

Liefertermine mit Wochenenden und Feiertagen
berücksichtigen

Um neben den Wochenenden auch Feiertage zu berücksichtigen, erweitern Sie die
ARBEITSTAG-Formel um einen Bereich, der Feiertage enthält.

Listen Sie dazu die Feiertage in einem Bereich auf; zum Beispiel im Bereich E1:E2.

Verwenden Sie folgende Formel:

=ARBEITSTAG(A1; B1; E1:E2)

Hierbei werden sowohl Wochenenden als auch die angegebenen Feiertage
übersprungen. Sie verlängern die Lieferzeiten entsprechend. Es werden nur die Tage
von Montag bis Freitag, die keine Feiertage sind, berücksichtigt.

Liefertermin berechnen mit Berücksichtigung von Wochenenden und Feiertagen

Liefertermine rückwärts berechnen

Möchten Sie ermitteln, wann ein Paket spätestens versandt werden muss, um
rechtzeitig anzukommen? Auch hierfür können Sie die Funktion ARBEITSTAG()
einsetzen.

Die Formel lautet:

=ARBEITSTAG(A1; -B1)

– A1 ist das gewünschte Lieferdatum.

– B1 gibt die Anzahl der Tage an, die zurückgerechnet werden.

Beispiel:

– Lieferdatum: 10.12.2024

– Lieferzeit: 15 Tage

– Das Ergebnis zeigt den spätesten Versandtermin: 19.11.2024.

Formel zur Rückwärtsberechnung von Lieferterminen

Mit der Funktion ARBEITSTAG() lassen sich Liefertermine flexibel und exakt
berechnen. Berücksichtigen Sie dabei Wochenenden und Feiertage, um realistische
Planungen zu erstellen.

Ablauf von Garantiefristen mit Excel
berechnen

Erfahren Sie, wie Sie Excel nutzen, um das Ende von Garantiefristen
oder Vertragslaufzeiten präzise zu berechnen. So stellen Sie sicher, dass
keine wichtigen Deadlines verpasst werden.

Zuletzt geändert am 18.03.2026

Besonders im Geschäftsalltag, aber auch im privaten Bereich, ist es wichtig, den
Überblick über Garantiefristen oder Vertragslaufzeiten zu behalten, um rechtzeitig
handeln zu können.

Dabei kann Excel ein nützliches Werkzeug sein. Mit einfachen Formeln und Funktionen
ermöglicht Excel eine schnelle Berechnung und Verwaltung von Garantiefristen.

Garantiefrist in Tagen

Besteht eine Garantiefrist in Tagen, dann können Sie den Garantieablauf berechnen,
indem Sie einfach die Anzahl der Tage zum Ausgangsdatum hinzuaddieren.

Beispiel: Sie kaufen einen Artikel am 15.01.2024 und erhalten eine Garantie oder ein
Rückgaberecht von 14 Tagen. Da ein Datum in Excel auch eine Zahl repräsentiert, die
lediglich in einem anderen Zahlenformat dargestellt wird, können Sie auf das Datum
einfach 14 Tage addieren.

Lassen Sie sich das Ergebnis allerdings im Zahlenformat Datum anzeigen, und Sie
erhalten den Tag im Format TT.MM.JJJJ dargestellt, an dem die Garantie abgelaufen ist
(siehe folgende Abbildung).

Tage zu einem Datum addieren

Sollte ein anderes Zahlenformat eingestellt sein, dann können Sie dies schnell ändern,
indem Sie die entsprechende Zelle markieren und anschließend die Tastenkombination
Strg + 1 drücken.

Es öffnet sich das Dialogfeld Zellen formatieren. Klicken Sie hier auf das Register 
Zahlen. Unter Kategorie wählen Sie Datum aus. Anschließend können Sie unter Typ
das gewünschte Datumsformat markieren. Bestätigen Sie Ihre Einstellung, indem Sie
das Dialogfeld durch einen Klick auf OK schließen.

Datumsformat festlegen

Garantiefrist in Monaten

Besteht die Garantiefrist in Monaten, dann können Sie hierfür die Funktion EDATUM()
einsetzen. Die Funktion EDATUM() in Excel wird verwendet, um ein Datum zu
berechnen, das eine bestimmte Anzahl von Monaten vor oder nach einem
Ausgangsdatum liegt.

EDATUM() besitzt die folgende Syntax:

=EDATUM(Ausgangsdatum; Monate)

– Ausgangsdatum: Das Datum, von dem aus gerechnet wird.

– Monate: Die Anzahl der Monate, die zum Ausgangsdatum addiert (positiver
Wert) oder subtrahiert (negativer Wert) werden sollen.

Soll die Garantiefrist nicht 14 Tage, sondern 24 Monate betragen, sehen Sie die
entsprechende Formel in der folgenden Abbildung.

Funktion EDATUM() für die Addition von Monaten

Garantiefrist in Jahren

Sollte die Garantiefrist in Jahren angegeben sein, dann können Sie hierfür auch die
Funktion EDATUM() einsetzen. Geben Sie hier im zweiten Argument die Anzahl der
Jahre ein und multiplizieren Sie diese mit 12, da ein Jahr aus 12 Monaten besteht.

Formel zur Berechnung einer Garantiezeit in Jahren

Ablauf von Verträgen, Abonnements oder
Mitgliedschaften

Mit dieser einfachen Funktion EDATUM() können Sie nicht nur den Ablauf einer
Garantiezeit berechnen. Sie können damit auch das Ende einer Vertragslaufzeit
ermitteln, das späteste Kündigungsdatum oder das Datum, zu dem Sie einen Vertrag
verlängern müssen.

Beispiel: Ihr Servicevertrag mit einem Dienstleister läuft über 12 Monate und
verlängert sich automatisch, wenn Sie diesen nicht 3 Monate vor Ablauf kündigen.
Damit Sie diese Frist nicht verpassen, soll das Datum angezeigt werden, zu dem Sie
den Vertrag überprüfen wollen – mit einem Puffer von sieben Tagen.

Die entsprechende Formel für das Prüfdatum (in Zelle B8) lautet:

=EDATUM(B1; (B2-B3)) - 1 - B4

In den folgenden Zellen stehen die entsprechenden Einträge:

– B1: Vertragsbeginn (Datum)

– B2: Vertragslaufzeit in Monaten

– B3: Frist zur Kündigung in Monaten

– B4: Puffer zur Prüfung und Kündigung in Tagen

Wenn Sie in einer weiteren Zeile eine Markierung anzeigen lassen wollen, ob das
aktuelle Datum das Prüfdatum erreicht hat, geben Sie die Formel ein:

=WENN(HEUTE() >= B8; "G"; "")

Dabei steht das Prüfdatum in Zelle B8. Die Zelle, mit dem Eintrag „G“, falls die Prüfung
„wahr“ ist, formatieren Sie in der Schriftart Wingdings, roter Schrift und fett. Dann wird
das Symbol der folgenden Abbildung angezeigt.

Vertragsverlängerung prüfen mit flexiblem Prüfdatum

IBAN mit Excel berechnen und überprüfen

Wie Sie mit einer Excel-Formel aus einer Kontonummer und Bankleitzahl
die dazugehörende IBAN berechnen. Entsprechend können Sie eine
IBAN auf Gültigkeit prüfen sowie Kontonummer und Bankleitzahl
ermitteln.

Zuletzt geändert am 18.03.2026

Die International Bank Account Number, kurz IBAN, ist eine standardisierte
internationale Kennung für Bankkonten. Sie ermöglicht es, Zahlungen
grenzüberschreitend sicher und effizient abzuwickeln.

In diesem Beitrag erfahren Sie, wie Sie mithilfe von Excel eine IBAN berechnen und
eine vorhandene IBAN auf ihre Gültigkeit prüfen.

Aufbau der IBAN

Die IBAN besteht in Deutschland aus 22 Zeichen, die sich wie folgt zusammensetzen:

Länderkennzeichen (2 Stellen)

Das Länderkennzeichen ist eine zweistellige Abkürzung, die das Land identifiziert, in
dem das Konto geführt wird. Für Deutschland ist dies beispielsweise DE.

Prüfziffer (2 Stellen)

Die Prüfziffer dient zur Plausibilitätsprüfung der IBAN. Sie wird anhand eines
mathematischen Verfahrens aus den anderen Zeichen der IBAN berechnet. Stimmt die
Prüfziffer nicht mit der berechneten Prüfziffer überein, ist die IBAN ungültig.

Basic Bank Account Number (BBAN)

Die BBAN ist die nationale Kontonummer, die sich aus der Bankleitzahl und der
Kontonummer zusammensetzt.

– Bankleitzahl (8 Stellen): Die Bankleitzahl identifiziert das Kreditinstitut, bei
dem das Konto geführt wird.

– Kontonummer (10 Stellen): Die Kontonummer ist die eindeutige
Kennzeichnung des Kontos innerhalb des Kreditinstituts.

Berechnung einer IBAN in Excel

Um eine IBAN aus einer Bankleitzahl und Kontonummer mithilfe von Excel zu
berechnen, können Sie auf verschiedene Funktionen und Formeln zurückgreifen und so
die Länderkennung, die Prüfziffer und die Kontonummer gemäß dem IBAN-Format des
jeweiligen Landes kombinieren.

Hierbei ist zu beachten, dass jedes Land unterschiedliche Regeln für die IBAN-Bildung
hat. Für eine deutsche IBAN können Sie hierbei die folgende Formel in Excel
verwenden:

="DE"&TEXT((98-REST((62 * (1+REST(B1; 97)) + 27 * REST(B2; 97)); 97));
"00")&B1&TEXT(B2; "0000000000")

Die Formel geht davon aus, dass die Bankleitzahl in der Zelle B1 und die Kontonummer
in der Zelle B2 steht.

IBAN mit Excel berechnen

Überprüfung einer vorhandenen IBAN auf
Richtigkeit

Sie können Excel auch einsetzen, um eine vorhandene IBAN auf Richtigkeit zu
überprüfen.

=B1="DE"&TEXT((98 - REST((62 * (1+REST(TEIL(B1; 5; 8); 97)) + 27 *
REST(RECHTS(B1; 10); 97)); 97)); "00")&TEIL(B1; 5; 8)&TEXT(RECHTS(B1;
10); "0000000000")

Die Formel geht davon aus, dass sich die zu überprüfende IBAN in der Zelle B1
befindet. Handelt es sich um eine rechnerisch richtige IBAN, dann wird als Ergebnis 
WAHR ausgegeben. Handelt es sich um eine ungültige IBAN, dann wird der FALSCH
als Ergebnis ausgegeben.

IBAN mit Excel prüfen

Bankleitzahl aus einer IBAN ableiten

Sie können aus einer IBAN auch eine Bankleitzahl ableiten. Hierfür können Sie die
Funktion TEIL() in Excel einsetzen.

Die Bankleitzahl kommt nach der Länderkennung (2 Zeichen) und der Prüfziffer
(2 Zeichen) und steht somit an fünfter Stelle. Da die Bankleitzahl immer 8 Ziffern lang
ist, können Sie somit die folgende Formel zum Ableiten der Bankleitzahl einsetzen:

=TEIL(B1; 5; 8)

Die Formel geht davon aus, dass sich die IBAN in der Zelle B1 befindet.

Bankleitzahl aus IBAN berechnen

Kontonummer aus einer IBAN ableiten

Analog zur Bankleitzahl können Sie auch die Kontonummer aus der IBAN ableiten.
Die Kontonummer kommt direkt nach der Bankleitzahl und ist in Deutschland 10
Ziffern lang. Sie beginnt mit dem 13. Zeichen in der IBAN.

Die Kontonummer können Sie mit der folgenden Formel ableiten:

=TEIL(B1; 4+8+1; 10)

Die Formel geht davon aus, dass sich die IBAN in der Zelle B1 befindet.

Kontonummer aus IBAN berechnen

Funktion LET() – mehr Übersicht in Excel

Mit der Excel-Funktion LET() vergeben Sie Rechenparametern
sprechende Namen. Mit diesen Namen können Sie Rechenformeln
definieren, die leicht lesbar und verständlich sind. Zwei Beispiele zeigen,
wie Sie LET() einsetzen.

Zuletzt geändert am 18.03.2026

Was leistet die Excel-Funktion LET()?

Die Excel-Funktion LET() ermöglicht Benutzern, komplexe Formeln zu erstellen und
gleichzeitig den Code lesbarer und wartbarer zu gestalten.

Die LET()-Funktion erlaubt die Definition von benannten Zwischenergebnissen
innerhalb einer Formel, was zu einer verbesserten Struktur und Verständlichkeit der
Formelcodes führt.

Die wesentlichen Vorteile der LET()-Funktion sind:

– Verbesserte Übersichtlichkeit: Die Funktion erlaubt die Zuweisung von
Namen zu einzelnen Werten oder ganzen Berechnungen in einer Formel. Dies
erleichtert die Lesbarkeit und Verständlichkeit von langen und komplexen
Formeln.

– Verbesserte Leistung: Durch die Definition von Variablen mit
wiederkehrenden Werten oder Berechnungen können diese in der gesamten
Formel wiederverwendet werden. Dadurch wird die Leistung verbessert, da
bestimmte Berechnungen nur einmal durchgeführt werden müssen.

So ist die LET-Funktion definiert

Die Syntax der LET()-Funktion lautet wie folgt:

=LET(Name1; Wert1; [Name2; Wert2]; …; Ausdruck)

– Name1, Name2 etc.: Die Namen, die den Zwischenergebnissen (Variablen)
zugewiesen werden sollen.

– Wert1, Wert2 etc.: Die Werte oder Ausdrücke, die den entsprechenden
Variablen zugewiesen werden sollen.

– Ausdruck: Der Hauptausdruck, der die berechneten Zwischenergebnisse
verwendet.

Ein einfaches Beispiel für die Excel-Funktion LET()

Das folgende Beispiel zeigt die Funktionsweise der LET()-Funktion. Aus einem
gegebenen Nettoumsatz und Steuersatz soll die Umsatzsteuer berechnet werden.

Beispiel für die Excel-Funktion LET()

In den ersten Argumenten definieren Sie zuerst die Namen (Variablen) und die
Herkunft des jeweiligen Wertes, die in der Funktion verwendet werden sollen:

– Nettobetrag =LET(Nettobetrag; A2; Steuersatz; B2;
Nettobetrag*Steuersatz)

– Steuersatz =LET(Nettobetrag; A2; Steuersatz; B2; Nettobetrag*Steuersatz)

Im letzten Argument der Funktion LET() wird dann immer die eigentliche Berechnung
durchgeführt. Hier werden die definierten Namen (Variablen) für die Berechnung
verwendet.

– Berechnung =LET(Nettobetrag; A2; Steuersatz; B2; 
Nettobetrag*Steuersatz)

Da die definierten Namen nur in der jeweiligen Formel gelten, können Sie die Funktion
daher ohne Probleme in andere Zellen kopieren, ohne dass es zu Namenskonflikten in
der Arbeitsmappe kommt.

Ein komplexes Beispiel und die Stärke der LET-
Funktion

Je komplexer die Formel oder Berechnung, desto mehr kann die Funktion LET() ihre
Stärke, bezogen auf die verbesserte Übersichtlichkeit, ausspielen.

Schauen Sie sich das folgende Beispiel an und versuchen Sie, die Formel in der
Spalte D schnell zu durchschauen.

Komplexer Anwendungsfall für LET()

Die Aufstellung enthält die Umsätze eines Konzerns nach Filialen, die in
unterschiedlichen Kontinenten ansässig sind. Der jeweilige Filialumsatz wird mit dem
Durchschnittsumsatz des Kontinents verglichen.

Die folgenden Ergebnisse werden dann wie folgt zurückgegeben:

– +, wenn der Umsatz größer als der Durchschnittsumsatz des entsprechenden
Kontinents ist

– -, wenn der Umsatz kleiner als der Durchschnittsumsatz des entsprechenden
Kontinents ist

– , wenn der Umsatz gleich dem Durchschnittsumsatz des entsprechenden
Kontinents ist

Hätten Sie dies sofort aus der verwendeten Formel herauslesen können?

=WENN(C2>MITTELWERTWENNS($C$2:$C$21; $A$2:$A$21; A2); "+";
WENN(C2<MITTELWERTWENNS($C$2:$C$21; $A$2:$A$21; A2); "-"; 0))

Vermutlich nicht! Schauen Sie sich die Alternative mit der Funktion LET() an:

Auswertung der Umsatzliste mit der LET-Funktion

Im ersten Teil der Funktion LET() werden zuerst die Namen (Variablen) und dann die
entsprechenden Werte definiert. Sie teilen der Funktion also mit:

Der Umsatz einer Filiale steht im Feld C2:

=LET(Umsatz; C2; Durchschnitt; MITTELWERTWENNS($C$2:$C$21; $A$2:$A$21; A2);
WENN(Umsatz>Durchschnitt; "+"; WENN(Umsatz<Durchschnitt; "-"; 0)))

Der Durchschnitt der Umsätze pro Kontinent wird mit der Formel
MITTELWERTWENNS() berechnet:

=LET(Umsatz; C2; Durchschnitt; MITTELWERTWENNS($C$2:$C$21;
$A$2:$A$21; A2); WENN(Umsatz>Durchschnitt; "+"; WENN(Umsatz<Durchschnitt;
"-"; 0)))

Am Schluss, im letzten Argument, erfolgt dann die eigentliche Berechnung mit den
definierten Namen (Variablen):

=LET(Umsatz; C2; Durchschnitt; MITTELWERTWENNS($C$2:$C$21; $A$2:$A$21; A2); 
WENN(Umsatz>Durchschnitt; "+"; WENN(Umsatz<Durchschnitt; "-"; 0)))

Die eigentliche Berechnung in der Funktion LET() ist deutlich übersichtlicher und
einfacher zu verstehen als die Ausgangsformel.

Weitere Vereinfachung der Formel mit LET()

Sie können das Ganze mit LET() gliedern und somit übersichtlicher gestalten, wenn Sie
auch die Argumente für die Funktion MITTELWERTWENNS() mit Namen in der Funktion
LET() versehen:

=LET(Regionbereich; $A$2:$A$21; Region; A2; Umsatzbereich; $C$2:$C$21;
Umsatz; C2; Durchschnitt; MITTELWERTWENNS(Umsatzbereich; Regionbereich;
Region); WENN(Umsatz>Durchschnitt; "+"; WENN(Umsatz<Durchschnitt; "-"; 0)))

Dieses Beispiel zeigt, dass auch Zellbereiche als Parameter definiert werden können: 
Regionbereich; $A$2:$A$21

Verkettung von Parametern mit der Funktion LET()

Hinweis: Nach ein paar Anwendungen definieren und lesen Sie viel schneller die
Rechenregel mit der LET-Funktion. Probieren Sie es aus!

Mit LAMBDA() eigene Funktionen in Excel
erstellen

Was die Lambda-Funktion in Excel leistet, wie sie aufgebaut und
angewendet wird. Damit lassen sich komplexe Berechnungen als eigene
Funktion definieren und einfach einsetzen. Mit einem Beispiel zur
Definition einer Lambda-Funktion.

Zuletzt geändert am 18.03.2026

LAMBDA ermöglicht es Benutzern von Microsoft 365, benutzerdefinierte Funktionen zu
erstellen, ohne auf VBA (Visual Basic for Applications) zurückgreifen zu
müssen.

Die LAMBDA-Funktion erweitert die Möglichkeiten von Excel, indem Sie individuelle
Funktionen definieren und in Formeln verwenden.

Was die LAMBDA-Funktion auszeichnet

Die LAMBDA-Funktion hat die folgenden Merkmale:

Benutzerdefinierte Funktionen erstellen: Mit LAMBDA() können Benutzer eigene
Funktionen definieren, indem sie eine Formel angeben, die einen oder mehrere
Parameter akzeptiert und einen Ergebniswert zurückgibt. Dies eröffnet eine breite
Palette von Anwendungsmöglichkeiten, da Benutzer maßgeschneiderte Funktionen für
spezifische Anforderungen selbst erstellen können.

Lesbarkeit und Strukturierung: Die Verwendung von LAMBDA() verbessert die
Lesbarkeit von Formeln und erleichtert die Strukturierung von komplexen
Berechnungen. Benutzer können komplexe Aufgaben in übersichtliche, leicht
verständliche Funktionen aufteilen, was die Pflege und Fehlerbehebung erheblich
erleichtert.

Wiederverwendbarkeit: Ein großer Vorteil von LAMBDA() besteht darin, dass einmal
erstellte Funktionen in anderen Formeln wiederverwendet werden können. Dies fördert
die Effizienz, da komplexe Berechnungen oder Prozesse in benutzerdefinierten
Funktionen verpackt und dann einfach in verschiedenen Teilen der Arbeitsmappe
wiederverwendet werden können.

Parameter und Argumente: LAMBDA-Funktionen können Parameter haben, denen
beim Aufruf Werte übergeben werden. Dies ermöglicht eine flexible Nutzung der
Funktionen, da unterschiedliche Eingabewerte zu verschiedenen Ergebnissen führen
können.

So ist die Lambda-Funktion definiert

Die LAMBDA-Funktion hat die folgende Syntax:

=LAMBDA(Parameter1; Parameter2; …; Ausdruck)

– Parameter1; Parameter2; …: Die Parameter, die die Funktion akzeptiert.
Sie können bis zu 253 Parameter definieren.

– Ausdruck: Der Ausdruck oder die Formel, die von der Funktion ausgeführt
wird. Dieser Ausdruck kann Verweise auf die Parameter enthalten. Der
Ausdruck ist immer das letzte Argument in der LAMBDA-Funktion.

Beispiel für den Einsatz der Lambda-Funktion

Ein praktisches Beispiel soll die Möglichkeiten mit der Lambda-Funktion verdeutlichen.
Aus einem gegebenen Bruttoumsatz und Steuersatz definieren Sie eine Funktion, die
den entsprechenden Nettoumsatz berechnet.

Namen definieren

Benutzerdefinierte Funktionen mit LAMBDA() werden über die Namen in Excel
definiert. Aktivieren Sie daher im Menüband die Befehlsfolge Registerkarte Formeln >
Befehlsgruppe Definierte Namen > Befehl Namen definieren.

Namen für Formeln definieren

Daraufhin öffnet sich das Dialogfeld Neuer Name. Vergeben Sie hier unter Name:
einen aussagekräftigen Namen für die benutzerdefinierte Funktion. Im Beispiel sei dies
der Name Nettoumsatz.

Lambda-Funktion festlegen

Unter Bezieht sich auf: kommt jetzt die LAMBDA-Funktion zum Einsatz:

=LAMBDA(Bruttoumsatz; Steuersatz; Bruttoumsatz/(1+Steuersatz))

Die ersten beiden Argumente sind die Parameter, die Sie für die
benutzerdefinierte Funktion Nettoumsatz definieren und für die eigentliche
Berechnung benötigen:

=LAMBDA(Bruttoumsatz; Steuersatz; Bruttoumsatz/(1+Steuersatz))

Im letzten Argument legen Sie die eigentliche Berechnung Ihrer benutzerdefinierten
Funktion fest

=LAMBDA(Bruttoumsatz; Steuersatz; Bruttoumsatz/(1+Steuersatz))

Bestätigen Sie Ihre benutzerdefinierte Funktion, indem Sie das Dialogfeld durch einen
Klick auf OK schließen.

Formel-Name durch Lambda() definieren

Lambda-Funktion anwenden

Erfassen Sie jetzt in einer Zelle den Ausdruck =netto, dann bekommen Sie alle
Funktionen in Excel vorgeschlagen, die mit NETTO beginnen. In der dargestellten Liste
der Funktionen befindet sich jetzt auch die neu mit LAMBDA() definierte Funktion 
Nettoumsatz().

Lambda-Funktionen erscheinen in der Funktionen-Liste

Doppelklicken Sie auf Nettoumsatz und Sie bekommen die definierten Argumente als
Eingabe-Argumente vorgeschlagen. Erfassen Sie den Bezug oder Wert zum
Bruttoumsatz in der Excel-Tabelle.

Parameter 1 festlegen

Erfassen Sie ein Semikolon, um zum nächsten Argument der benutzerdefinierten
Funktion zu gelangen. Erfassen Sie nun einen Bezug oder Wert für den Steuersatz.

Parameter 2 festlegen

Bestätigen Sie die Formel und schließen Sie die Eingabe durch Drücken der Enter-
Taste ab. Sie erhalten daraufhin den entsprechenden Nettoumsatz aus dem
Bruttoumsatz berechnet.

Ergebnis der Funktion Lambda() = Nettoumsatz()

Hinweis: Die von Ihnen definierte Lambda-Funktion steht in der Excel-Mappe (Datei)
zur Verfügung, in der Sie diese definiert haben.

Komplexe Berechnungen mit Lambda()
vereinfachen

Der wesentliche Vorteil der Lambda-Funktion ergibt sich dann, wenn der Teil 
Ausdruck, der die Rechenregel enthält, komplex ist und Sie die Berechnung an
unterschiedlichen Stellen Ihrer Excel-Mappe durchführen wollen.

Beispiel: Sie wollen für ein Jahr und eine Kalenderwoche das Datum des Montags in
der entsprechenden Woche berechnen. Die Parameter der Lambda-Funktion sind
dann:

– Parameter 1 = Jahr

– Parameter 2 = Kalenderwoche

Zur Berechnung des Datums für den Montag benötigen Sie folgende Formel:

=LAMBDA(JAHR; KALENDERWOCHE; ("4.1."&JAHR) +
KALENDERWOCHE*7-7-REST("2.1."&JAHR; 7))

Beispiel für eine komplexe Lambda-Formel

Zufällige Kennwörter durch Excel erstellen
lassen

Mit Excel lassen sich nicht nur Zufallszahlen erzeugen, sondern auch
zufällige Buchstaben. Wird die entsprechende Funktion kombiniert und
verkettet, entstehen damit Passwörter oder zufällige Folgen von
Buchstaben und Ziffern.

Zuletzt geändert am 18.03.2026

Das Generieren von Kennwörtern kann manchmal nervig und zeitraubend sein. Lassen
Sie doch Excel schnell und einfach ein Kennwort für Sie erstellen.

Mit der folgenden Formel können Sie ein zufälliges Kennwort aus Groß- und
Kleinbuchstaben erzeugen:

=ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,5) * 32 + 64,5;
0))

Durch die Funktion Zufallszahl erzeugen Sie hier eine Zufallszahl, die anschließend
nach einigen mathematischen Umrechnungen durch die Funktion ZEICHEN() in einen
Groß- oder Kleinbuchstaben umgerechnet wird.

Wichtig: Bei der Funktion ZUFALLSZAHL() handelt es sich um eine volatile Funktion,
die bei jedem Rechenvorgang automatisch eine neue Zufallszahl und damit einen
neuen Buchstaben erzeugt. Der Buchstabe oder Ihr Kennwort ändert sich deshalb
jedes Mal, wenn Sie in Ihrer Excel-Tabelle eine neue Aktion durchführen.

Damit Sie ein 8-stelliges Kennwort erstellen können, müssen Sie diese Formel 8 Mal
hintereinander verketten. Zum Verketten können Sie das &-Zeichen verwenden.

=ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,5) * 32 + 64,5; 0))&
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,5) * 32 + 64,5; 0))&
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,5) * 32 + 64,5; 0))&

ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,5) * 32 + 64,5; 0))&
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,5) * 32 + 64,5; 0))&
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,5) * 32 + 64,5; 0))&
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,5) * 32 + 64,5; 0))&
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,5) * 32 + 64,5; 0))

Wenn Sie neben Groß- und Kleinbuchstaben auch Ziffern zufällig im Kennwort
integrieren wollen, dann verwenden Sie die folgende Formel:

=WENN(ZUFALLSZAHL()<0,33; ABRUNDEN(ZUFALLSZAHL() * 10; 0);
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,66) * 32 + 64,5;
0)))

Und hier die Formel, wenn es ein 8-stelliges Kennwort aus großen und kleinen
Buchstaben und aus Zahlen sein soll:

=WENN(ZUFALLSZAHL()<0,33; ABRUNDEN(ZUFALLSZAHL() * 10; 0);
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,66) * 32 + 64,5; 0)))&
WENN(ZUFALLSZAHL()<0,33; ABRUNDEN(ZUFALLSZAHL() * 10; 0);
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,66) * 32 + 64,5; 0)))&
WENN(ZUFALLSZAHL()<0,33; ABRUNDEN(ZUFALLSZAHL() * 10; 0);
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,66) * 32 + 64,5; 0)))&
WENN(ZUFALLSZAHL()<0,33; ABRUNDEN(ZUFALLSZAHL() * 10; 0);
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,66) * 32 + 64,5; 0)))&
WENN(ZUFALLSZAHL()<0,33; ABRUNDEN(ZUFALLSZAHL() * 10; 0);
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,66) * 32 + 64,5; 0)))&
WENN(ZUFALLSZAHL()<0,33; ABRUNDEN(ZUFALLSZAHL() * 10; 0);
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,66) * 32 + 64,5; 0)))&
WENN(ZUFALLSZAHL()<0,33; ABRUNDEN(ZUFALLSZAHL() * 10; 0);
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,66) * 32 + 64,5; 0)))&
WENN(ZUFALLSZAHL()<0,33; ABRUNDEN(ZUFALLSZAHL() * 10; 0);
ZEICHEN(RUNDEN(ZUFALLSZAHL() * 26 + (ZUFALLSZAHL()>0,66) * 32 + 64,5; 0)))

Empfehlungen aus dem Management-
Handbuch

Tabellen bearbeiten in Excel
Wie Sie Tabellen für die Datenanalyse vorbereiten und die gewünschten Bereiche
auswählen, ergänzen oder kürzen. Mit einer Anleitung für die besondere Excel-
Funktion BEREICH.VERSCHIEBEN und Funktionen, um Listenwerte zu erzeugen.
https://www.business-wissen.de/id/kapitel/317/

Zählfunktionen in Excel
Datenbestände werden in Excel mit Zählfunktionen unter die Lupe genommen, um
Fehler zu erkennen oder erste Besonderheiten zu ermitteln. Das sind die wichtigen
Zählfunktionen und ihre Möglichkeiten für die Datenanalyse.
https://www.business-wissen.de/id/kapitel/314/

Textfunktionen in Excel
Wie Sie spezielle Textfunktionen in Excel nutzen, um Text aus Zellen auszulesen,
aufzuteilen und besser zu verstehen.
https://www.business-wissen.de/id/kapitel/309/

Nutzen Sie als

Premium-Mitglied
alle

Handbuch-Kapitel
mit mehr als

3.000 Checklisten und Excel-Vorlagen

Jetzt anmelden

www.business-wissen.de/anmelden/

